# $Id: PKGBUILD 179637 2013-03-08 23:40:40Z foutrelis $
# Maintainer: Thomas Bächler <thomas@archlinux.org>

pkgname=libnl
pkgver=3.2.21
pkgrel=1
pkgdesc="Library for applications dealing with netlink sockets"
arch=(i686 x86_64)
url="http://www.infradead.org/~tgr/libnl/"
license=(GPL)
depends=(glibc)
backup=(etc/libnl/classid etc/libnl/pktloc)
options=(!libtool)
source=("$url/files/$pkgname-$pkgver.tar.gz")
sha256sums=('2dc80f043116ec8610d9d418f5f5e9283f3d9a3a48b8d51a8cb0146b3f279604')

build() {
  cd "$srcdir/$pkgname-$pkgver"
  ./configure \
    --prefix=/usr \
    --sysconfdir=/etc \
    --sbindir=/usr/bin \
    --disable-static
  make
}

package() {
  cd "$srcdir/$pkgname-$pkgver"
  make DESTDIR="$pkgdir" install
}
